// Nunchuk_Test - Arduino IDE 1.0.3 + UNO - F_Zia - 2013_0210
#include "Wire.h"
uint8_t  nunchuk_buf[6];     // store six bytes data from nunchuk
uint8_t  TxStatus, count;    // transmission status, data counter
void nunchuk_init()

{

  Wire.begin();

  delay(1);

  Wire.beginTransmission(0x52);  // transmit to device 0x52
  Wire.write((uint8_t)0xF0);     // 1st initialisation register
  Wire.write((uint8_t)0x55);     // 1st initialisation value
  TxStatus = Wire.endTransmission();

  if (TxStatus==0) Serial.println("Nunchuk Initialization OK"); 

     else          Serial.println("Nunchuk Initialization ERROR");

  Wire.beginTransmission(0x52);  // transmit to device 0x52
  Wire.write((uint8_t)0xFB);     // 2nd initialisation register
  Wire.write((uint8_t)0x00);     // 2nd initialisation value
  TxStatus = Wire.endTransmission();

  if (TxStatus==0) Serial.println("Nunchuk Initialization OK"); 

     else          Serial.println("Nunchuk Initialization ERROR");

}

void nunchuk_request_data()

{

  Wire.beginTransmission(0x52);  // transmit to device 0x52
  Wire.write((uint8_t)0x00);     // request data from nunchuk
  TxStatus = Wire.endTransmission();

  if (TxStatus!=0) Serial.println("Nunchuk Request Data   ERROR"); 

  delay(1);

}

void nunchuk_get_data()

{

  count=0;

  Wire.requestFrom (0x52, 6);         // get six bytes from nunchuk
  while (Wire.available()){

    nunchuk_buf[count] = Wire.read(); // receive byte as an integer
    count++;

  }

}

void nunchuk_print_data()

{

  for (int i=0; i < count; i++){      // print data values
    Serial.print(nunchuk_buf[i]);     // in serial monitor
    Serial.print('\t');               // seperated by tabs
  }

  Serial.println();                   // move cursor to newline
}

void setup() 

{

  Serial.begin(9600);

  nunchuk_init();

}

void loop() 

{

  nunchuk_request_data();

  nunchuk_get_data();

  // mask 6th data byte to reveal C,Z switch values
  nunchuk_buf[5] = nunchuk_buf[5] & 0b00000011;

  nunchuk_print_data();

  delay(500);  

}

